

με παραδείγματα:

```
{ Resonz.ar(WhiteNoise.ar(0.5), MouseX.kr(1000,1800), 0.05) }.play  
{ Resonz.ar(WhiteNoise.ar(0.5), 2000, MouseX.kr(0.001,0.01), MouseY.kr(1, 5))  
}.play
```

Η Ringz λειτουργεί όπως η Resonz με τη διαφορά πως αντί για τον συντελεστή Q , χρησιμοποιεί έναν συντελεστή χρόνου μείωσης του σήματος κατά -60 dB. Η Ringz χρησιμοποιείται ως μονάδα Formant στην κλάση Klank. Συντάσσεται ως εξής:

```
Ringz.ar (in: 0, freq: 440, decaytime: 1, mul: 1, add: 0)
```

in	Το σήμα εισόδου.
freq	Η συχνότητα συνήχησης.
decaytime	Ο χρόνος μείωσης κατά -60 dB.
mul	Πολλαπλασιαστικός συντελεστής επί των τιμών της εξόδου.
add	Αθροιστικός συντελεστής επί των τιμών της εξόδου (dc offset).

με παράδειγμα:

```
{ Ringz.ar(WhiteNoise.ar(0.005), 2000, 0.5) }.play
```

Η κλάση Formlet αναπαράγει ένα συνηχητικό φίλτρο στο οποίο, αντί της συνηθούς συμμετρίας της καμπύλης απόκρισης που παρουσιάζουν τα φίλτρα διέλευσης εκατέρωθεν της κεντρικής συχνότητας, μπορούν να προσδιοριστούν διαφορετικοί χρόνοι ανόδου (*attack time*) και χρόνοι καθόδου (*decay time*) του φίλτρου. Αλγοριθμικά είναι ισοδύναμο με:

```
Ringz(in, freq, decaytime) - Ringz(in, freq, attacktime)
```

Συντάσσεται ως εξής:

```
Formlet.ar(in: 0, freq: 440, attacktime: 1, decaytime: 1, mul: 1, add: 0)
```

in	Το σήμα εισόδου.
freq	Η συχνότητα συνήχησης.
attacktime	Ο χρόνος ανόδου ενίσχυσης κατά $+60$ dB.
decaytime	Ο χρόνος καθόδου μείωσης κατά -60 dB.
mul	Πολλαπλασιαστικός συντελεστής επί των τιμών της εξόδου.
add	Αθροιστικός συντελεστής επί των τιμών της εξόδου (dc offset).

με χαρακτηριστικό παράδειγμα:

```
{ Formlet.ar(WhiteNoise.ar(0.005), 2000, 0.5, 0.1) }.play;
```


4.3. Φασματικές, Μη-Γραμμικές Τεχνικές: Διαμόρφωση Πλάτους (AM), Ring Modulation, Διαμόρφωση Συχνότητας (FM).

Με τον όρο διαμόρφωση εννοούμε την επέμβαση σε κάποιο από τα χαρακτηριστικά της ταλάντωσης ενός σήματος (πλάτος, συχνότητα, φάση) με βάση τα χαρακτηριστικά μιας άλλης ταλάντωσης. Το σήμα το οποίο υφίσταται τη διαμόρφωση ονομάζεται *φορέας (carrier)*, ενώ το σήμα το οποίο επεμβαίνει στα

χαρακτηριστικά του φορέα *διαμορφωτής (modulator)*. Οι τεχνικές διαμόρφωσης αποτελούν φασματικές μη-γραμμικές τεχνικές καθώς εισάγουν μη προ-υπάρχουσα φασματική πληροφορία. Δύο βασικές τεχνικές διαμόρφωσης είναι η Διαμόρφωση Πλάτους και η Διαμόρφωση Συχνότητας, καθώς και η τεχνική της Διαμόρφωσης Δακτυλίου (Ring Modulation) η οποία αποτελεί ειδική περίπτωση της Διαμόρφωσης Πλάτους.


4.3.1. Διαμόρφωση Πλάτους (Amplitude Modulation).

Κατά τη Διαμόρφωση Πλάτους μια ημιτονοειδής ταλάντωση (σήμα διαμορφωτή) διαμορφώνει τις τιμές πλάτους του σήματος φορέα (Σχήμα 4.10).



Σχήμα 4.10 Διάγραμμα Διαμόρφωσης Πλάτους.

Κατ' αυτόν τον τρόπο το πλάτος ταλάντωσης του φορέα AMP μεταβάλλεται διαρκώς με βάση την ημιτονοειδή ταλάντωση που παράγεται από τον διαμορφωτή. Οι τιμές δηλαδή πλάτους μεταβάλλονται περιοδικά σε ένα εύρος τιμών $[AMP+a, AMP-a]$, με συχνότητα περιοδικότητας ίση προς τη συχνότητα του διαμορφωτή FREQ_m (Σχήμα 4.11).




Σχήμα 4.11 Εάν οι τιμές που παράγονται από μια ταλάντωση (i) αθροιστούν σε μια σταθερή τιμή πλάτους AMP (ii), τότε θα προκύψει μια διαρκώς μεταβαλλόμενη τιμή πλάτους ταλάντωσης AMP' (iii). Εάν στη συνέχεια η τιμή αυτή χρησιμοποιηθεί σαν πλάτος σε μια δεύτερη ταλάντωση, τότε η παραγόμενη κυματομορφή θα εμφανίζεται να μεταβάλλει διαρκώς το πλάτος της σε ένα εύρος τιμών $[AMP-a, AMP+a]$ με περίοδο μεταβολής ίση προς την περίοδο της αρχικής ταλάντωσης $1/FREQ_m$ (iv).

Σε αρκετούς αλγόριθμους διαμόρφωσης το πλάτος a του διαμορφωτή συνηθίζεται να δίνεται ως ποσοστό επί της τιμής AMP του φορέα. Ισχύει δηλαδή:


$$a = AMP \times i$$

όπου i το ποσοστό διαμόρφωσης εκφρασμένο σε δεκαδική τιμή. Έτσι για $i = 0$ δεν παρουσιάζεται καθόλου διαμόρφωση αφού το γινόμενο $AMP \times i$ στην περίπτωση αυτή ισούται με μηδέν. Αντίστοιχα για ποσοστό 100%, δηλαδή $i = 1$, έχουμε πλήρη διαμόρφωση. Ο δείκτης αυτός i , καλείται και *συντελεστής διαμόρφωσης* (*index of modulation*) (Σχήμα 4.12).



Σχήμα 4.12 Αλγόριθμος Διαμόρφωσης Πλάτους με την τιμή πλάτους του διαμορφωτή εκφρασμένη ως ποσοστό επί του πλάτους φορέα AMP (συντελεστής διαμόρφωσης i).

Το παραγόμενο φάσμα αποτελείται από τη συχνότητα του φορέα καθώς και δύο πλευρικές εκατέρωθεν συχνότητες σε απόσταση ίση προς τη συχνότητα του διαμορφωτή με πλάτος ίσο προς $AMP \times i/2$ (Σχήμα 4.13).



Σχήμα 4.13 Παραγόμενο φάσμα από την κλασική Διαμόρφωση Πλάτους.

Έτσι, για χαμηλές τιμές συχνότητας του διαμορφωτή, δηλαδή από 1 έως περίπου 15 Hz, παράγεται το φαινόμενο του tremolo. Για μεγάλες μάλιστα τιμές του συντελεστή διαμόρφωσης παράγονται και διακριθηματικά φαινόμενα (Σχήμα 4.14):

```

/**tremolo effect (modulating frequencies between 1 and 15 Hz)
(
SynthDef("am-tremolo", { arg out=0, c_freq, c_amp, main=0.2;
var mod, car, index;

index = MouseY.kr(0,1); //index of modulation
mod = SinOsc.ar(MouseX.kr(1,15), 0, index*c_amp); //modulator
car = SinOsc.ar(c_freq, 0, mod + c_amp); //carrier
Out.ar(out, [1,1]*car*main);
}
).send(s);
)

```


```

s.sendMsg("/s_new", "am-tremolo", 1001, 0, 1, "c_freq", 100, "c_amp", 0.2);

//different carrier frequencies
s.sendMsg("/n_set", 1001, "c_freq", 200);
s.sendMsg("/n_set", 1001, "c_freq", 400);
s.sendMsg("/n_set", 1001, "c_freq", 1000);
s.sendMsg("/n_set", 1001, "c_freq", 4000);

s.sendMsg("/n_free", 1001);

```



Σχήμα 4.14 Παραγωγή διακροτηματικών φαινομένων με συντελεστή διαμόρφωσης στο 100% ($i = 1$).

Καθώς η τιμή συχνότητας του διαμορφωτή καθορίζει και την απόσταση των παραγόμενων πλευρικών συχνοτήτων από τη κεντρική συχνότητα του φορέα (Σχήμα 4.13), για υψηλότερες τιμές συχνότητας του διαμορφωτή, από δηλαδή περίπου 20 Hz και άνω, παράγεται νέο φασματικό περιεχόμενο με ακουστικά διακριτές πλευρικές συχνότητες:

```

/***/ separating side frequencies (modulating frequencies between 20 and 1500 Hz)
(
SynthDef("am-2", { arg out=0, c_freq, c_amp, main=0.2;
var mod, car, index;

index = MouseY.kr(0,1); // index of modulation
mod = SinOsc.ar(MouseX.kr(20,1500), 0, index*c_amp); //modulator
car = SinOsc.ar(c_freq, 0, mod + c_amp); //carrier
Out.ar(out, [1,1]*car*main);
}
).send(s);
)

s.sendMsg("/s_new", "am-2", 1002, 0, 1, "c_freq", 100, "c_amp", 0.2);

//different carrier frequencies

```


```

s.sendMessage("/n_set", 1002, "c_freq", 200);
s.sendMessage("/n_set", 1002, "c_freq", 400);
s.sendMessage("/n_set", 1002, "c_freq", 1000);
s.sendMessage("/n_set", 1002, "c_freq", 4000);

s.sendMessage("/n_free", 1002);


```

Σε πολλές περιπτώσεις σχεδίασης αλγορίθμων Διαμόρφωσης Πλάτους το σήμα διαμόρφωσης παρουσιάζεται ως *μονοπολικό* (*unipolar*) [Roads, 1999]. Σε αντίθεση με το *διπολικό σήμα* (*bipolar*) του οποίου οι τιμές ταλάντωσης κυμαίνονται μεταξύ -1 και 1, το μονοπολικό σήμα ταλαντώνεται μεταξύ 0 και 1 (Σχήμα 4.15).



Σχήμα 4.15 Διπολικό (άνω) και μονοπολικό (κάτω) σήμα ταλάντωσης.

Αυτού του τύπου οι κυματομορφές είναι ιδιαίτερα χρήσιμες σε περιπτώσεις που χρησιμοποιούνται για να περιγράψουν φαινόμενα τα οποία αφορούν *μεγάλη ρυθμίσεων* στον ήχο και δεν μπορούν να περιέχουν αρνητικές τιμές. Τέτοια παραδείγματα αποτελούν η κίνηση ενός δείκτη θέσεων μνήμης σε πίνακες, buffer κ.λπ., μια περιβάλλουσα πλάτους, η κίνηση μιας ηχητικής πηγής σε ένα στερεοφωνικό πεδίο (*ranning*) κ.ο.κ. Στην περίπτωση που η ταλάντωση του διαμορφωτή αφορά μονοπολικό σήμα, η μονοπολικότητα διασφαλίζει πως κατά την ελάχιστη τιμή του διαμορφωτή (τιμή πλάτους 0) το σήμα του φορέα δε θα παρουσιάζει διαμόρφωση, ενώ κατά τη μέγιστη τιμή του διαμορφωτή (πλάτος a) το σήμα φορέα διαμορφώνεται πλήρως, παρουσιάζοντας έτσι μια κανονικότητα. Προκειμένου να υλοποιηθεί κάτι τέτοιο αρκεί το σήμα διαμόρφωσης να έχει *αθροιστική μετατόπιση* των τιμών του (*DC-offset*) και, επακόλουθα, οι τιμές που παράγονται να διαιρεθούν κατά το ήμισυ (Σχήμα 4.16).




Σχήμα 4.16 Διαδικασία μετατροπής του διπολικού (i) σε μονοπολικό (v) σήμα ταλάντωσης, με αθροιστική μετατόπιση κατά a (ii και iii) και επακόλουθη ελάττωση των τιμών κατά το ήμισυ (iv και v).

Στο περιβάλλον του SuperCollider όλες οι μονάδες γεννητριών παρέχουν ένα όρισμα `add` το οποίο αντιστοιχεί στην τιμή αθροιστικής μετατόπισης των παραγόμενων τιμών τους. Ειδικότερα, στην κλάση `SinOsc` το όρισμα αυτό απαντάται τέταρτο στη σειρά ορισμάτων της.

```
(
SynthDef("am-unipolar", { arg out=0, c_freq, c_amp, main=0.2;
var mod, car, index, mod_amp;
index = MouseY.kr(0,1);
mod_amp = index*c_amp;
mod = SinOsc.ar(MouseX.kr(1,15), 0, mod_amp, mod_amp); //modulator-unipolar
car = SinOsc.ar(c_freq, 0, mod/2 + c_amp); //carrier
Out.ar(out, [1,1]*car*main);
}
).send(s);
)
```


4.3.2. Πολλαπλασιασμός Σημάτων (Ring Modulation).

Μια παραλλαγή της Διαμόρφωσης Πλάτους αποτελεί η τεχνική της Διαμόρφωσης Δακτυλίου (Ring Modulation), κατά την οποία οι έξοδοι δύο ταλαντωτών πολλαπλασιάζονται μεταξύ τους (Σχήμα 4.17).



Σχήμα 4.17 Αλγόριθμος πολλαπλασιασμού σημάτων (Ring Modulation).

Στην πιο απλή του μορφή ο αλγόριθμος υλοποιείται από δύο ημιτονοειδείς. Το παραγόμενο φάσμα μοιάζει σημαντικά με αυτό της Διαμόρφωσης Πλάτους. Εμφανίζει δηλαδή τις δύο πλευρικές συχνότητες στα σημεία $(FREQ1-FREQ2)$ και $(FREQ1+FREQ2)$, χωρίς όμως να εμφανίζει κεντρική συχνότητα φορέα (Σχήμα 4.18).




Σχήμα 4.18 Παραγόμενο φάσμα από τον πολλαπλασιασμό δύο ημιτονοειδών. Η συχνότητα φορέα f_c απουσιάζει.

Με υλοποίηση στο περιβάλλον του SuperCollider:

```
(
SynthDef("rm", { arg out=0, amp1=0.5, amp2=0.5, main=0.7;
var mod1, mod2, sig;
mod1 = SinOsc.ar(MouseX.kr(3,30), 0, amp1); //source-1: sinewave
mod2 = SinOsc.ar(MouseY.kr(30,1000), 0, amp2); //source-2: sinewave
sig = mod1*mod2;
Out.ar(out, [1,1]*sig*main);
}
).send(s);
)

s.sendMsg("/s_new", "rm", 1001, 0, 1);
s.sendMsg("/n_free", 1001);
```

Άλλες υλοποιήσεις της τεχνικής αυτής αφορούν τον πολλαπλασιασμό ημιτονοειδούς με ένα σύνθετο σήμα. Στην περίπτωση αυτή για κάθε μερική συχνότητα $f_{c1}, f_{c2}, \dots, f_{cn}$ του σύνθετου σήματος παράγεται και ένα νέο ζευγάρι τιμών $[f_{cn} - f_m, f_{cn} + f_m]$ εκατέρωθεν τους (Σχήμα 4.19).




Σχήμα 4.19 Παραγόμενο φασματικό περιεχόμενο από τον πολλαπλασιασμό μιας απλής ημιτονοειδούς f_m με σύνθετη κυματομορφή η οποία αποτελείται από αρμονικούς $f_{c1}, f_{c2}, f_{c3}, f_{c4}$, κ.ο.κ.

Ιδιαίτερο ενδιαφέρον έχει η τεχνική αυτή όταν το ένα σήμα αποτελείται από παλμούς χαμηλής συχνότητας, καθώς αυτοί ουσιαστικά επενεργούν ως διακοπτόμενες περιβάλλουσες δυναμικής επί του δεύτερου σήματος:

```
(
SynthDef("rm-2", { arg out=0, freq2;
var mod1, mod2, sig;
mod1 = Pulse.ar(MouseX.kr(1,20),0.5, 0.2); //source-1: low frequency pulses
mod2 = SinOsc.ar(freq2, 0, MouseY.kr(0,1)); //source-2: sinewave
sig = mod1*mod2;
Out.ar(out, [1,1]*sig);
}
).send(s);
)
s.sendMsg("/s_new", "rm-2", 1002, 0, 1, "freq2", 140);
s.sendMsg("/n_set", 1002, "freq2", 440);
s.sendMsg("/n_set", 1002, "freq2", 4400);

s.sendMsg("/n_free", 1002);
```

Μια τρίτη περίπτωση, τέλος, αφορά και τα δύο σήματα να είναι σύνθετα. Τότε οι μερικές του ενός σήματος μετατοπίζονται εκατέρωθεν σε πολλαπλάσια αποστάσεων $f_{m1}, f_{m2}, f_{m3}, \dots$ των θέσεων των μερικών του άλλου σήματος (Σχήμα 4.20).



Σχήμα 4.20 Πολλαπλασιασμός δύο σύνθετων σημάτων αποτελούμενοι από αρμονικούς. Οι αρμονικοί f_{m1}, f_{m2} και f_{m3} του ενός σήματος μετατοπίζονται εκατέρωθεν των αρμονικών f_{c1}, f_{c2}, f_{c3} κ.ο.κ του άλλου, σε σύνολα προσθαιρέσεων $[f_{c1}-f_{m1}, f_{c1}-f_{m2}, f_{c1}-f_{m3}]$, $[f_{c1}+f_{m1}, f_{c1}+f_{m2}, f_{c1}+f_{m3}]$ κ.λπ.


Ακολουθεί ένα παράδειγμα όπου στο περιβάλλον του SuperCollider παλμοί χαμηλής συχνότητας διαμορφώνουν έναν στιγμιαίο παλμό (κλάση Impulse), πηγή ιδιαίτερα πλούσια σε πλήθος αρμονικών, με αποτέλεσμα την παραγωγή ενός ιδιαίτερου ηχοχρώματος:

```
(
SynthDef("rm-3", { arg out=0, main=0.5;
var mod1, mod2, sig;
mod1 = Pulse.ar(MouseX.kr(1,20),0.5, 0.5); //source-1: low frequency pulses
mod2 = Impulse.ar(MouseY.kr(1,1000), 0, 0.5); //source-2: impulse
sig = mod1*mod2;
Out.ar(out, [1,1]*sig*main);
}
).send(s);
)

s.sendMsg("/s_new", "rm-3", 1003, 0, 1);
s.sendMsg("/n_free", 1003);
```


4.3.3. Διαμόρφωση Συχνότητας (Frequency Modulation).

Κατά τη Διαμόρφωση Συχνότητας η συχνότητα ενός ταλαντωτή φορέα (carrier) διαμορφώνεται με βάση τις τιμές που παράγει ένας ταλαντωτής διαμόρφωσης (modulator). Οι τιμές του διαμορφωτή αθροίζονται με μια σταθερή τιμή συχνότητας FREQ πριν καταλήξουν στον ταλαντωτή φορέα (Σχήμα 4.21).




Σχήμα 4.21 Αλγόριθμος της Διαμόρφωσης Συχνότητας.

Κατ' αυτόν τον τρόπο το σήμα που παράγεται στην έξοδο έχει περιοδικώς μεταβαλλόμενη συχνότητα με εύρος τιμών $[FREQ-a, FREQ+a]$. Η περιοδικότητα αυτή ισοδυναμεί με την περίοδο του διαμορφωτή (Σχήμα 4.22).



Σχήμα 4.22 Κατά τη Διαμόρφωση Συχνότητας η τιμή συχνότητας του φορέα $FREQ$ (ii) μεταβάλλεται με βάση τις τιμές ταλάντωσης του φορέα (i). Οι νέες τιμές $FREQ'$ (iii) παράγουν ένα σήμα με περιοδικά μεταβαλλόμενη συχνότητα και περίοδο μεταβολής ίση με την περίοδο του διαμορφωτή $1/FREQ_m$ (iv).

Το φάσμα το οποίο παράγεται αποτελείται από ένα σύνθετο σήμα το οποίο περιέχει τη συχνότητα του φορέα καθώς και ένα πλήθος πλευρικών συχνοτήτων σε αποστάσεις ακέραιων πολλαπλασίων της f_m εκατέρωθεν της συχνότητας φορέα, σε ζεύγη δηλαδή τιμών $[f_c + (n \times f_m), f_c - (n \times f_m)]$ όπου n το ακέραιο πολλαπλάσιο. Οι πλευρικές συχνότητες εκτείνονται χονδρικά σε μια έκταση ισοδύναμη με το εύρος τιμών που παράγει ο ταλαντωτής διαμόρφωσης, τιμή δηλαδή ίση προς $2a$ (Σχήμα 4.23). Το βασικό λοιπόν χαρακτηριστικό το οποίο καθορίζει τον βαθμό διαμόρφωσης είναι το πλάτος με το οποίο ταλαντώνεται ο ταλαντωτής διαμόρφωσης, χαρακτηριστικό το οποίο καλείται και *βάθος διαμόρφωσης (depth of modulation)*. Αυξάνοντας το πλάτος του διαμορφωτή αυξάνεται αντίστοιχα και το βάθος διαμόρφωσης, ενώ ταυτόχρονα μεγαλώνει και το πλήθος παράπλευρων συχνοτήτων το οποίο παράγεται από τη διαμόρφωση.



Σχήμα 4.23 Παραγόμενο φάσμα κατά τη Διαμόρφωση Συχνότητας.


Για τον παραπάνω λόγο οι τιμές πλάτους του διαμορφωτή σε αρκετούς αλγόριθμους Διαμόρφωσης Συχνότητας δίδονται σε Hz. Η σχέση μάλιστα αυτή αποτυπώνεται καλύτερα με τον συντελεστή διαμόρφωσης i , όπου:

$$i = a / fm$$

ή:


$$a = i \times fm$$

Η παραπάνω σχέση φανερώνει πως το παραγόμενο φάσμα εκτείνεται σε πολλαπλάσια i της συχνότητας διαμόρφωσης fm , το πλήθος των οποίων σχετίζεται με το πλάτος a του διαμορφωτή. Έτσι, αυξάνοντας τον συντελεστή i αυξάνεται και το πλήθος των πλευρικών παραγόμενων συχνοτήτων του φάσματος. Σε αρκετούς αλγορίθμους μάλιστα το πλάτος του διαμορφωτή συνηθίζεται να διατυπώνεται με βάση την παραπάνω σχέση (Σχήμα 4.24).



Σχήμα 4.24 Ο αλγόριθμος Διαμόρφωσης Συχνότητας με το πλάτος του διαμορφωτή εκφρασμένο ως γινόμενο του συντελεστή διαμόρφωσης επί της συχνότητας διαμορφωτή.

Τα χαρακτηριστικά των πλευρικών συχνοτήτων, οι τιμές δηλαδή πλάτους και φάσης τους, καθορίζονται με ακρίβεια από μια ομάδα μαθηματικών συναρτήσεων, οι συναρτήσεις Bessel 1ου τύπου (Εικόνα 4.6). Οι συναρτήσεις αυτές συμβολίζονται ως $J_n(x)$, όπου n ο βαθμός ή η τάξη (order) της συνάρτησης.



Εικόνα 4.6 Απεικόνιση τριών διαφορετικών βαθμών των συναρτήσεων Bessel 1ου τύπου: του μηδενικού βαθμού με κόκκινο χρώμα, του πρώτου με πράσινο και του δεύτερου βαθμού με μπλε χρώμα (πηγή: [https://commons.wikimedia.org/wiki/File:Bessel_Functions_\(1st_Kind,_n%3D0,1,2\).svg_-_media/File:Bessel_Functions_\(1st_Kind,_n%3D0,1,2\).svg](https://commons.wikimedia.org/wiki/File:Bessel_Functions_(1st_Kind,_n%3D0,1,2).svg_-_media/File:Bessel_Functions_(1st_Kind,_n%3D0,1,2).svg)).

Αναλόγως τον βαθμό της, κάθε τέτοια συνάρτηση περιγράφει τις τιμές πλάτους και φάσης της αντίστοιχης πλευρικής συχνότητας του παραγόμενου φάσματος. Έτσι, η καμπύλη συνάρτησης 1ου βαθμού $J_1(x)$ περιγράφει το πλάτος και τη φάση του πρώτου ζεύγους πλευρικών συχνοτήτων, δηλαδή των $f_c - f_m$ και $f_c + f_m$, η καμπύλη συνάρτησης 2ου βαθμού $J_2(x)$ περιγράφει τις τιμές πλάτους και φάσης του δεύτερου ζεύγους πλευρικών συχνοτήτων, δηλαδή των $f_c - 2f_m$ και $f_c + 2f_m$ κ.ο.κ. Η καμπύλη μηδενικού βαθμού $J_0(x)$ περιγράφει το πλάτος και τη φάση με την οποία θα εμφανιστεί η συχνότητα του φορέα f_c . Όλες οι παραπάνω πληροφορίες συναρτώνται με την τιμή του συντελεστή διαμόρφωσης i , ο οποίος αποτυπώνεται στον άξονα x του διαγράμματος. Αναλόγως λοιπόν την τιμή του i στον άξονα x , λαμβάνουμε την αντίστοιχη τιμή στον άξονα y ως πληροφορία της τιμής πλάτους και φάσης από κάθε καμπύλη για κάθε μια πλευρική συνιστώσα. Η πληροφορία της φάσης κάθε συνιστώσας συνάγεται εμμέσως από την τιμή του πλάτους: εάν από το διάγραμμα προκύπτει αρνητική τιμή πλάτους, όπως φαίνεται για παράδειγμα για $i = 5$ στις συναρτήσεις μηδενικού και πρώτου βαθμού, ισχύει ως τιμή πλάτους η απόλυτη τιμή και φάση ίση προς 180° . Για τις συχνότητες της φθίνουσας πλευράς ισχύει μάλιστα πως ανά περιττό πολλαπλάσιο η πληροφορία της φάσης αναστρέφεται. Ισχύει δηλαδή:

Ζεύγος Πλευρικών Συνιστωσών	Φθίνουσα Πλευρά			Αύξουσα Πλευρά		
	συχνότητα	πλάτος	φάση	συχνότητα	πλάτος	φάση
1ο	$f_c - f_m$	$J_1(i)$	180°	$J_1(i)$	$f_c - f_m$	0
2ο	$f_c - 2f_m$	$J_2(i)$	0	$J_2(i)$	$f_c - 2f_m$	0
3ο	$f_c - 3f_m$	$J_3(i)$	180°	$J_3(i)$	$f_c - 3f_m$	0
4ο	$f_c - 4f_m$	$J_4(i)$	0	$J_4(i)$	$f_c - 4f_m$	0
5ο	$f_c - 5f_m$	$J_5(i)$	180°	$J_5(i)$	$f_c - 5f_m$	0

Πίνακας 4.1 Τιμές συχνότητας, πλάτους και φάσης για τα πρώτα πέντε παραγόμενα ζεύγη πλευρικών συνιστωσών κατά τη Διαμόρφωση Συχνότητας.

Για πολύ χαμηλές συχνότητες διαμόρφωσης της τάξης του 1 έως περίπου 15 Hz και κατ' αντιστοιχία μικρές τιμές συντελεστή διαμόρφωσης, από δηλαδή περίπου 1 έως 10, παράγεται το φαινόμενο του *vibrato*:

```
(
SynthDef("fm-vibrato", { arg out=0, c_freq, c_amp, main=0.2;
var mod, car, m_freq;
m_freq = MouseX.kr(1,15); //modulation frequency
mod = SinOsc.ar(m_freq, 0, MouseY.kr(0,10)*m_freq); //modulator
car = SinOsc.ar(c_freq+mod, 0, c_amp); //carrier
Out.ar(out, [1,1]*car*main);
}
).send(s);
)

s.sendMsg("/s_new", "fm-vibrato", 1001, 0, 1, "c_freq", 100, "c_amp", 0.2);
//different carrier frequencies
s.sendMsg("/n_set", 1001, "c_freq", 200);
s.sendMsg("/n_set", 1001, "c_freq", 400);
s.sendMsg("/n_set", 1001, "c_freq", 1000);
s.sendMsg("/n_set", 1001, "c_freq", 4000);

s.sendMsg("/n_free", 1001);
)
```

Για μεγαλύτερες τιμές συχνότητας διαμόρφωσης και συντελεστή διαμόρφωσης, παράγεται ένα ιδιαίτερα πλούσιο φασματικό περιεχόμενο. Καθώς μάλιστα οι πλευρικές συχνότητες μεγαλώνουν δημιουργείται πλήθος αναδιπλώσεων στο συχνοτικό φάσμα με πολύ ενδιαφέροντα ηχοχρωματικά χαρακτηριστικά.

```
(
SynthDef("fm-2", { arg out=0, c_freq, c_amp, main=0.2;
var mod, car, m_freq;
m_freq = MouseX.kr(20,1500); //modulation frequency
mod = SinOsc.ar(m_freq, 0, MouseY.kr(0,100)*m_freq); //modulator
car = SinOsc.ar(c_freq+mod, 0, c_amp); //carrier
Out.ar(out, [1,1]*car*main);
}
).send(s);
)

s.sendMsg("/s_new", "fm-2", 1002, 0, 1, "c_freq", 100, "c_amp", 0.2);
//different carrier frequencies
s.sendMsg("/n_set", 1002, "c_freq", 200);
s.sendMsg("/n_set", 1002, "c_freq", 400);
s.sendMsg("/n_set", 1002, "c_freq", 1000);
s.sendMsg("/n_set", 1002, "c_freq", 4000);

s.sendMsg("/n_free", 1002);
)
```

4.4. Χρονικές, Γραμμικές Τεχνικές: Σύνθεση Πινάκων Κυματομορφής, Σειρές Καθυστέρησης.

Κατά τις χρονικές-γραμμικές τεχνικές σύνθεσης οι αλγόριθμοι Σύνθεσης Ήχου επεμβαίνουν στα δείγματα ήχου από τα οποία αποτελείται το αρχικό σήμα ενώ ταυτόχρονα δεν προσθέτουν επιπλέον φασματικό περιεχόμενο. Οι επεμβάσεις αυτές στα δείγματα του αρχικού σήματος αφορούν τρεις τύπους: α) χρονικής καθυστέρησης των δειγμάτων, β) κλιμάκωσης δειγμάτων και γ) χρονικής καθυστέρησης με ταυτόχρονη κλιμάκωση. Η Σύνθεση Πίνακα Κυματομορφής αποτελεί μια τεχνική κλιμάκωσης δειγμάτων, καθώς οι τιμές από τις οποίες αποτελείται ένας πίνακας μπορούν να αυξηθούν ή να μειωθούν πολλαπλασιαστικά